
ABSTRACT  

Imputation of data is often a critical step in compensating for missing values when analysis is due to be 
performed. The need for effective imputation has been demonstrated within the Sentinel Initiative, a 
program sponsored by the U.S. Food and Drug Administration (FDA) created to monitor the safety of 
FDA-regulated medical products via an active surveillance system. Since Sentinel functions as a 
distributed data network, data completeness can vary, emphasizing the importance of quality data 
imputation. The focus of this paper is on comparison of imputed values, in particular those reporting race, 
ethnicity, sex, and gender.  

While current data partners are primarily using the Bayesian Improved Surname and Geocoding (BISG) 
method and a standardized imputation method developed by RTI International (RTI) for imputation, the 
tools and approaches available to compare reported versus imputed values vary much more widely. This 
paper describes and discusses the use of PROC COMPARE, implementation of crosstabulation 
particularly using PROC FREQ, and PROC MEANS, application of hash methods as well as regular 
expressions, and PROC SQL. Testing of these methods was done on fabricated descriptive reporting 
datasets which emulate those aggregated across the Sentinel population. This paper includes a 
comparative analysis of the methods to evaluate their effectiveness based on a range of priorities and 
relevant study outcomes.  

INTRODUCTION  

Imputation of data is performed across disciplines as part of data management and analysis. Once 
imputed values are established, it makes sense to compare them to the originals to understand how often 
the process is introducing change and in what contexts. This paper explores the impact of imputation 
within the FDA's Sentinel Initiative, an active surveillance of regulated medical products once they are on 
the market. However, the goal of comparison extends far beyond the pharmaceutical or clinical realms, 
and the techniques explored here are applicable across all data types. Criteria for picking techniques is 
included with discussion of each of the procedures along with sample code that incorporates descriptive 
comments and sample output.  

BACKGROUND 

Sentinel refers to an active surveillance system sponsored by the FDA to monitor the safety of FDA-
regulated medical products.  Administrative and clinical data are collected from various healthcare 
organizations, known as Data Partners (DPs), and maintained in a distributed data environment. While 
DPs must standardize their data to a Sentinel Common Data Model (SCDM) and store them as SAS 
datasets, they always maintain ownership of the usage and transfer of this data.  In the distributed data 
environment, the Sentinel Operations Center (SOC) distributes analytic programs that the DPs run locally 
behind their firewalls, and aggregated results are sent back to the SOC. Patient privacy is always 
preserved.  

A major area of concern in health studies is understanding how conditions, treatments, and outcomes 
impact different segments of the population. However, reliable sources of demographic data are needed 
to explore possible distinctions. Though efforts to collect this data in a consistent manner have improved 
over time, there is no one set way to do this. While the Sentinel Common Data Model (SCDM) 
standardizes demographic information like sex, race, and ethnicity within Sentinel, that does not 
compensate for values that are missing from the start. A supplied value of race and/or ethnicity may need 
to be compared to other factors related to the person like where they live or their surname.   

This creates some of the many reasons reliable imputation may be necessary. The first consideration is 
likely to be the completion of missing values in order to increase the number of valid records which can 
be included in a study. Closely linked to this is the public health priority of working to ensure appropriate 
representation of any demographic groups that are present within a given population. In parallel with 
having complete data, there remains the question of how accurate the imputation of the demographic 
values is, and even when using the well-established algorithms discussed later, the comparison of 
reported vs. imputed values is an important part of understanding how the data can be used. While this 
discussion is based on race and ethnicity, the questions it raises could apply to any data where 
imputation is being used.  

Examining health outcomes across different racial and ethnic groups can aid in identifying and addressing 
health disparities: preventable factors that disproportionally affect certain groups. Oftentimes, it is difficult 
to ascertain these health disparities due to limited or missing race and ethnicity data as most of this data 
is self-reported. To mitigate this, RAND corporation developed an algorithm that is widely used to impute 
these values, the Bayesian Improved Surname Geocoding (BISG) method. 

RTI International, a nonprofit research institute, developed an imputation algorithm to aid in the 
identification of Hispanic and Asian/Pacific Islander (A/PI) individuals by drawing on additional information 
available from the Centers for Medicare and Medicaid Services (CMS) and the U.S. Census Bureau. 
Though the limitations still present within this imputation method were acknowledged, even at the time of 
its development it more than doubled the sensitivity of Hispanic identification and increased the sensitivity 
of A/PI identification by nearly 45%. Thus, the RTI imputation method remains a standard when imputing 
race and ethnicity. 

More detailed information on these methods of imputation can be found in the links in the References 
section. 

INPUT DATASET FOR COMPARISON 

To illustrate the methods, the same source data, dem_l3_catvars, is being used in each case. The 
dataset below represents the same format as the aggregated demographic datasets provided by various 
DPs for purposes of quality assurance and data characterization, but all values are fabricated for 
purposes of this demonstration; see the Appendix for the code used to create this data. Below find a 
representative sample of the raw data in the dataset, dem_l3_catvars, showing variable names, a variety 
of variable values across fabricated DPs, and description of variable values names and values. 

 

DP: arbitrary Data Partner  

Count: number of people with combinations of 
reported values, imputed values, and sex 

hispanic: reported Hispanic; values include 
Y(yes), N(no), U(unknown) 

imputedhispanic: imputed Hispanic; values 
same as above 

race: reported race; values include 0 (unknown), 
1 (American Indian or Alaska Native), 2(Asian), 
3(Black or African American), 4(Native Hawaiian 
or other Pacific Islander), 5(White), M (Multi-
racial) 

imputedrace: imputed race; values same as 
above 

sex: reported sex; values include M(male), 
F(female), U(unknown)

Output 1. Sample of raw data 

METHODS OF COMPARISON 

PROC COMPARE 

When comparing variables in SAS, whether for reported vs. imputed values or another scenario, the most 
obvious choice may be to use PROC COMPARE.  Here the variable comparison is within the same 
dataset, so this is being done with the WITH statement: 

 

title1 'Proc Compare of Reported vs. Imputed Variables' ; 

 

/*SPECIFY NAME OF DATASET CONTAINING VARIABLES TO COMPARE*/ 

PROC COMPARE DATA = dem_l3_catvars; 

  VAR race hispanic /*SPECIFY REPORTED VARIABLES*/ ; 

  WITH imputedrace imputedhispanic /*SPECIFY IMPUTED VARIABLES*/ ; 

RUN; 

 

Output 2. Sample of PROC COMPARE default output  

 
The output shown in Output 2 is not very useful to analyze, and by default, SAS only prints the first 50 
differences for each comparison. PROC COMPARE has an option to produce an output dataset, and 
next, additional options are added to view a more thorough comparison.  The code below produces a 
SAS dataset, in addition to the output in the SAS results viewer shown in Output 2. 
 
/*SPECIFY DATASET CONTAINING VARIABLES TO COMPARE, AND OUTPUT DATASET*/  

/*SPECIFY OPTIONS FOR OUTPUT*/ 

/*OUTBASE: INCLUDES VALUES FROM THE BASE (REPORTED) VARIABLE*/ 

/*OUTCOMP: INCLUDES VALUES FROM THE COMPARISON (IMPUTED) VARIABLE*/ 

/*OUTDIF: INCLUDES STRING SHOWING DIFFERENCE BETWEEN THE COMPARED VARIABLES*/ 

PROC COMPARE DATA = dem_l3_catvars OUT = comp OUTBASE OUTCOMP OUTDIF ; 

  VAR race hispanic ; 

  WITH imputedrace imputedhispanic ; 

RUN ; 

Output 3. Sample of PROC COMPARE dataset output with options OUTBASE OUTCOMP OUTDIF 

The dataset output shows a vertical comparison of reported and imputed variables and displays an ‘X’ in 
the DIF rows signifying when there is a mismatch with the imputed value (see highlighted values).  

PROC COMPARE also allows for stratification by categorical variables if the dataset is sorted by said 
categorical variables. See the code below:   

 

/*SORT DATASET CONTAINING VARIABLES TO COMPARE BY THE CATEGORICAL VALUES*/ 

PROC SORT DATA = dem_l3_catvars ;   

  BY dp sex ; 

RUN ; 

 

title1 'Proc Compare of Reported vs. Imputed Variables (Stratified by DP and 

Sex)'; 

 

PROC COMPARE data = dem_l3_catvars OUT = comp_strat OUTBASE OUTCOMP OUTDIF ; 

  /*ADD BY STATEMENT WITH CATEGORICAL VARIABLES FOR STRATIFICATION*/ 

  BY dp sex ;  

  VAR race hispanic ; 

  WITH imputedrace imputedhispanic ; 

RUN ; 

Output 4. Sample of PROC COMPARE dataset output stratified by categorical variables with 
options OUTBASE OUTCOMP OUTDIF 

 
The output is lengthy in the results viewer window as up to 50 values of each stratification mismatch are 
displayed, but the dataset output as seen in Output 4 shows a clear vertical comparison of reported vs 
imputed variable value matches and mismatches stratified. Again, the highlighted ‘X’ values in the ‘DIF’ 
rows represent mismatches. For an additional analysis on this output dataset, PROC FREQ can be used 
to count the number of mismatches, or PROC COMPARE could be skipped entirely, going straight to 
using PROC FREQ for a comparison using crosstabulation. 
 

PROC FREQ: USING CROSSTABULATION 

A basic yet simple way to compare reported variables to their imputed counterpart is via PROC FREQ. 
The proc statement includes the specification of the dataset with the relevant variables and table 
statements to define the crosstabulations to be generated. Consider using the MISSING statement so 
that records without a value can be counted and compared. 

title1 "PROC FREQ WITH ALL COUNTS AND PERCENTS" ; 

 

/*SPECIFY DATASET THAT CONTAINS REPORTED AND IMPUTED VARIABLES*/ 

PROC FREQ DATA = dem_l3_catvars ; 

  /*USE TABLE STATEMENT TO SPECIFY SPECIFIC CROSSTABULATION*/ 

  /*USE OF "MISSING" INCLUDES ANY MISSING IN THE TABULATION*/ 

  /*OTHERWISE MISSINGS ARE COUNTED SEPARATELY*/ 

  TABLE race * imputedrace / MISSING ; 

  /*MULTIPLE TABLE STATEMENTS CAN BE INCLUDED IN ONE PROC*/ 

  TABLE hispanic * imputedhispanic / MISSING ; 

RUN ; 
 

Using this baseline implementation of PROC FREQ generates output in this format: 

 

Output 5. Sample of PROC FREQ default output with MISSING included 

However, when making these comparisons the default output of PROC FREQ is often too cluttered, as 
the standard includes not only the counts (frequencies) of each crosstabulated category but also the 
overall percent, the percent based on the row, and the percent based on the column. To suppress the 
percents, use the desired mix of the NOCOL, NOROW, and NOPERCENT options on the table 
statement. 

title1 "PROC FREQ WITH CUMULATIVE COUNTS AND PERCENTS REMOVED" ; 

 

PROC FREQ DATA = dem_l3_catvars ; 

 /*NOCOL SUPPRESSES COLUMN PERCENTAGE*/ 

 /*NOROW SUPPRESSES ROW PERCENTAGE*/ 

 /*NOPERCENT SUPPRESSES OVERALL PERCENTAGE*/ 

 TABLE race * imputedrace / MISSING NOCOL NOROW NOPERCENT ; 

 TABLE hispanic * imputedhispanic / MISSING NOCOL NOROW NOPERCENT ; 

RUN ; 

 

Removing the percents generates a cleaner comparison for data where typically the frequencies are the 
focus. 

 

 

 

Output 6. PROC FREQ output with MISSING NOCOL NOROW NOPERCENT included 

The TABLE statement can also be set up to run multiple crosstabulations of a variable against a list of 
other variables. This allows for comparisons between reported and imputed to be shown alongside 
comparisons to any other relevant variable or even to facilitate multiple comparisons if multiple methods 
of imputation were implemented and then need to be evaluated. 

 

title1 "PROC FREQ WITH ONE VARIABLE CROSSTABULATED WITH MULTIPLE VARIABLES" ; 

 

PROC FREQ DATA = dem_l3_catvars ; 

  /*PUTTING MULTIPLE VARIABLES IN PARENTHESES ALLOWS ONE VARIABLE*/ 

  /* TO BE CROSSTABULATED WITH EACH ONE IN THE SPECIFIED LIST    */ 

  TABLE race * ( imputedrace hispanic imputedhispanic ) / MISSING NOCOL NOROW  

NOPERCENT ; 

RUN ; 

This produces a series of crosstabulations from just one table statement. Particularly for large scale 
comparisons, this type of coding efficiency can be very helpful. 

 

 

 

Output 6. PROC FREQ output with one variable crosstabulated against multiple variables 

This approach also allows a categorical variable to be applied to multiple variables of interest. For this 
sample data, there are multiple data partners, and a reported or imputed variable could be counted for 
comparison between those data partners. 

title1 "PROC FREQ WITH CATEGORICAL VARIABLE USED TO CROSSTABULATE" ; 

 

PROC FREQ DATA = dem_l3_catvars ; 

  /*COUNTING BY ADDITIONAL CATEGORICAL VARIABLE*/ 

  TABLE dp * ( race imputedrace hispanic imputedhispanic ) /  

MISSING NOCOL NOROW NOPERCENT ; 

RUN ; 

Here is the output; note the similarities to the prior example: 

 

 

 

 Output 7. PROC FREQ output with crosstabulated by categorical variable 

Fun note: To remove the default title of “The FREQ Procedure”, include the following ODS statement: 

ODS NOPROCTITLE ; 

 

In selecting a method of crosstabulation, PROC TABULATE is also an option if a numeric variable is also 
available for counting relevant records. Such a “counter” variable could always be added to the data, but 
this discussion focuses on methods that can be applied to the data as it is. 

PROC MEANS 

Another method of comparison is possible through PROC MEANS. This proc is advantageous because it 
works well with character variables and provides counts that are easy to understand. Various forms of 
crosstabulation are also straightforward to build. Continuing to include the MISSING option allows 
comparisons where a value is not available to be included and to keep counts of observations the same. 

The proc statements specify the dataset with additional statements to specify particular variables, 
crosstabulations, and output datasets with the results of the proc. 

 

/* DATASET WITH REPORTED AND IMPUTED VARIABLES WITHOUT COUNT OR DP*/ 

PROC MEANS data = imputed ; 

  /*USE CLASS STATEMENT TO SPECIFY VARIABLES OF INTEREST*/ 

  CLASS race imputedrace / MISSING ; 

  /*USE TYPES STATEMENT TO SPECIFY CROSSTABULATIONS*/ 

  TYPES () race*imputedrace ; 

  /*USE OUTPUT STATEMENT TO SPECIFY OUTPUT DATASET*/ 

  OUTPUT OUT = race_imputedrace ; 

RUN ; 

The basic output from the proc includes the following: 

 

 

Output 8. PROC MEANS default output 

The output starts with the overall count of records and then displays the requested crosstabulations. The 
value of the first variable – in this case RACE – will not be repeated in each row, but instead is left blank 
after the first occurrence until the value changes. These are displayed in sorted order as are the values of 
the variable being crossed with the first with counts given for each pairing.  This output in Output 8 is 
partially truncated but would continue with all values of RACE crossed against IMPUTEDRACE. 

Since other work might need to be done with these values, producing an output dataset is often useful. 
The output data set can be limited to observations with the highest values if the NWAY option is included, 
but the goal here is to see everything that is produced first; consider whether NWAY would be useful 
once see all the output. 

 

 

 

Output 9. PROC MEANS output dataset 

The first line of the table is once again the total overall records and then each combination of the 
variables being crossed are included in sorted order. The final column, _FREQ_, displays the counts of 
the combinations. The automatically generated variable, _TYPE_, may be of use and is worth 
understanding. PROC MEANS defaults _TYPE_ to a value of 0 for the overall, and then if variables are 
being counted individually, the first variable would have a _TYPE_ value of 1 while the second would 
have a value of 2, etc. In this example, two variables are being crosstabulated, so the proc assigns a 
value of 3 to _TYPE_, indicating a third level or step after the existence of the two variables involved in 
the crosstabulation.  If you need to select certain categories of counts or are working with even more 
extensive crosstabulations, understanding what the value of _TYPE_ indicates can facilitate that 
subsetting. 

Further breakdown of the counts is possible with the use of a BY statement.  Variables should be sorted 
prior to PROC MEANS on the basis of the variables to be used in the BY statement. 

PROC MEANS DATA = imputed_sortbysex ; 

  /* USE BY STATEMENT TO SUBSET COUNTS OR CROSSTABULATION      */ 

  BY sex ; 

  CLASS race imputedrace / MISSING ; 

  TYPES () race * imputedrace ; 

  OUTPUT OUT = race_imputedrace_sex ; 

RUN ; 

Output 10. PROC MEANS output with BY statement 

In this example, the data is split by values of SEX before being counted and crosstabulated. 

As before, the overall count for the given category is provided first followed by the crosstabulation of the 
requested variables as they appear under that category.  

The output dataset follows a similar structure, but now has the BY statement variable as its first column 
followed by the variables being crosstabulated in its second and third.  Values of _TYPE_ and _FREQ_ 
follow the same format and have the same meaning as in the original version. 

Output 11. PROC MEANS with BY statement output dataset 

The same code could be applied to HISPANIC and IMPUTEDHISPANIC and would work in the same 
way. Since the output would be of the same format, it is being omitted, and just the code is included here. 

 

PROC MEANS DATA = imputed ; 

  CLASS hispanic imputedhispanic / MISSING ; 

  TYPES () Hispanic * imputedhispanic ; 

  OUTPUT OUT = hispanic_imputedhispanic ; 

RUN ; 

 

PROC MEANS DATA = imputed_sortbysex ; 

  BY sex ; 

  CLASS hispanic imputedhispanic / MISSING ; 

  TYPES () hispanic * imputedhispanic ; 

  OUTPUT OUT = Hispanic_imputedhispanic_sex ; 

RUN ; 

 

PROC SQL 

Another method to consider reported vs. imputed variable disparities is PROC SQL, which can be used 
for a variety of analyses, but here the comparison is done using counts and with a mismatch analysis. 
Unlike other procedures explored here, PROC SQL automatically includes records with missing values 
without an additional option to do so being required. 

See the code below for a basic crosstabulation: 

 

title1 'Proc Sql: Counts of Reported vs. Imputed Variables'; 

 

PROC SQL; 

  SELECT race 

        , imputedrace 

        , count(*) AS count 

  FROM dem_l3_catvars 

  GROUP BY race, imputedrace ; 

 

  SELECT hispanic 

        , imputedhispanic 

        , count(*) AS count 

  FROM dem_l3_catvars 

  GROUP BY hispanic, imputedhispanic ; 

QUIT; 

Output 12. PROC SQL default output with counts 

Further enhance this output with stratification by categorical variables. The beauty of PROC SQL is that it 
does not need to be sorted to produce sorted output, so this can be easily done by modifying the code as 
follows:   

title1 'Proc Sql: Counts of Reported vs. Imputed Variables (Stratified by DP 

and Sex)'; 

 

PROC SQL ; 

  SELECT dp 

        , sex 

        , race 

        , imputedrace 

        , count(*) AS count 

  FROM dem_l3_catvars 

  GROUP BY dp, sex, race, imputedrace 

  ORDER BY dp, sex ; 
 

  SELECT dp 

        , sex 

        , hispanic 

        , imputedhispanic 

        , count(*) AS count 

  FROM dem_l3_catvars 

  GROUP BY dp, sex, hispanic, imputedhispanic 

  ORDER BY dp, sex ; 

QUIT ; 

The output is as such:  

 

Output 13. PROC SQL output with stratifications 

If you only want to view counts of mismatched values, you can add a where statement to the PROC SQL 
code as such: 

title1 'Proc Sql: Counts of Reported vs. Imputed Variable Mismatches 

(Stratified by DP and Sex)'; 

PROC SQL ; 

  SELECT dp 

        , sex 

        , race 

        , imputedrace 

        , count(*) AS count 

  FROM dem_l3_catvars 

  WHERE race ne imputedrace 

  GROUP BY dp, sex, race, imputedrace 

  ORDER BY dp, sex ; 

 

  SELECT dp 

        , sex 

        , hispanic 

        , imputedhispanic 

        , count(*) AS count 

  FROM dem_l3_catvars 

  WHERE hispanic ne imputedhispanic 

  GROUP BY dp, sex, hispanic, imputedhispanic 

  ORDER BY dp, sex ; 

QUIT ; 

 

Output:  

 

Output 14. PROC SQL output with mismatch analysis 

Take this mismatch analysis further by producing both counts and percentages across one table.  

See the code below: 

 

 

 

 

 



17 

 

title1 'Proc Sql: Mismatch Analysis of Reported vs. Imputed Variables 

(Stratified by DP and Sex)'; 

PROC SQL; 

  SELECT dp 

        , sex 

        , SUM ( CASE WHEN race = imputedrace THEN 1 ELSE 0 END ) AS 

race_match 

        , sum( CASE WHEN race ne imputedrace THEN 1 ELSE 0 END ) AS 

race_mismatch 

        , sum( CASE WHEN hispanic = imputedhispanic THEN 1 ELSE 0 END ) AS 

hisp_match 

        , sum( CASE WHEN hispanic ne imputedhispanic THEN 1 ELSE 0 END ) AS 

hisp_mismatch 

        , count(*) AS total_count 

        , ( calculated race_match / calculated total_count) * 100 AS 

race_match_percent 

        , ( calculated race_mismatch / calculated total_count) * 100 AS 

race_mismatch_percent 

        , ( calculated hisp_match / calculated total_count) * 100 AS 

hisp_match_percent 

        , ( calculated hisp_mismatch / calculated total_count) * 100 AS 

hisp_mismatch_percent 

  FROM dem_l3_catvars 

  GROUP BY dp, sex ; 

QUIT ; 

Here values of 1 or 0 are assigned based on whether the variable comparison is a respective match or a 
mismatch, summing those values, and then dividing by the total count to obtain the percentages.  

The output is as follows: 

 

Output 15. PROC SQL output with mismatch analysis 

 

HASH TABLES 

When working with a large dataset, it is often important to consider efficiency, and using a hash table to 
compare variable disparities could make the difference. The examples here use hash tables to create 
summary counts of matching and mismatched values. Please note that because there are several lines of 
code required for the hash table comparisons, only race vs. imputedrace is displayed below. The code 
will be the same for the hispanic vs imputedhispanic variables so this code can easily be put in a macro to 
run comparisons for any reported vs imputed variables. 

DATA _NULL_ ( DROP = rc ) ; 

  /*INITIALIZE HASH OBJECT*/ 

  IF _n_ = 1 THEN DO ; 

    /*DEFINE HASH TABLE FOR RACE VS IMPUTEDRACE COMPARISON*/ 

    DECLARE HASH race_hash() ; 

    /*DEFINE KEY VARIABLES FOR COMPARISON*/ 

    race_hash.definekey ( 'race' , 'imputedrace' ) ; 

    /*DEFINE VARIABLES TO BE STORED IN HASH TABLE, INCLUDING KEY AND COUNT 

VARIABLES*/ 

    race_hash.definedata( 'race' , 'imputedrace' , 'race_match_count' , 

'race_mismatch_count' ) ; 

    race_hash.definedone() ; 

  END ; 

 

  /*READ OBSERVATIONS FROM INPUT DATASET AND DEFINE END OF DATASET AS EOF*/ 

  SET dem_l3_catvars END = eof; 

 

  IF race = imputedrace THEN DO ; /*CHECK FOR RACE MATCH AND COUNT MATCHES*/ 

    rc = race_hash.find() ; 

    IF rc ne 0 THEN DO ; 

      race_match_count = 0 ; 

      race_mismatch_count = 0 ; 

    END ; 

      race_match_count + 1; 

      rc = race_hash.replace(); 

  END ; 

  ELSE DO ; /*CHECK FOR RACE AND COUNT MISMATCHES*/ 

    rc = race_hash.find() ; 

    IF rc ne 0 THEN DO ; 

      race_match_count = 0 ; 

      race_mismatch_count = 0 ; 

    END ; 

    race_mismatch_count + 1 ; 

    rc = race_hash.replace() ; 

  END; 

 

  /*OUTPUT HASH SUMMARY TABLE AT END OF DATASET*/ 

  IF eof THEN DO ; 

    race_hash.output(dataset: 'race_summary' ) ; 

  END ; 

RUN ; 

 

/*SORT OUTPUT DATASET FOR READABILITY*/ 

PROC SORT DATA = race_summary ; 

  BY race imputedrace ; 

RUN ; 

The hash table code produces the dataset race_summary, as shown below: 

 

 

Output 16. HASH output dataset 

 

Take this a step further and consider stratifying these counts by the categorical variables DP and sex. 
Easily add these categorical variables when declaring the hash object as shown in the code below.  

DATA _NULL_ ( DROP = rc ) ; 

  /*INITIALIZE HASH OBJECT*/ 

  IF _n_ = 1 THEN DO ; 

    /*DEFINE HASH TABLE FOR RACE VS IMPUTEDRACE COMPARISON*/ 

    DECLARE HASH race_hash() ; 

    /*DEFINE KEY VARIABLES FOR STRATIFICATION AND COMPARISON*/ 

    race_hash.definekey ( 'dp' , 'sex' ,'race' , 'imputedrace' ) ; 

    /*DEFINE VARIABLES TO BE STORED IN HASH TABLE, INCLUDING KEY AND COUNT 

VARIABLES*/ 

    race_hash.definedata( 'dp' , 'sex' , 'race' , 'imputedrace' , 

'race_match_count' , 'race_mismatch_count' ) ; 

 race_hash.definedone() ; 

  END ; 

 

  /*READ OBSERVATIONS FROM YOUR INPUT DATASET AND DEFINE END OF DATASET AS 

EOF*/ 

  SET dem_l3_catvars END = eof; 

 

  IF race = imputedrace THEN DO ; /*CHECK FOR RACE MATCH AND COUTN MATCHES*/ 

    rc = race_hash.find() ; 

    IF rc ne 0 THEN DO ; 

      race_match_count = 0 ; 

      race_mismatch_count = 0 ; 

    END ; 

      race_match_count + 1; 

      rc = race_hash.replace(); 

  END ; 

  ELSE DO ; /*CHECK FOR RACE AND COUNT MISMATCHES*/ 

    rc = race_hash.find() ; 

    IF rc ne 0 THEN DO ; 

      race_match_count = 0 ; 

      race_mismatch_count = 0 ; 

    END ; 

    race_mismatch_count + 1 ; 

    rc = race_hash.replace() ; 

  END; 

 

  /*OUTPUT HASH SUMMARY TABLE AT END OF DATASET*/ 

  IF eof THEN DO ; 

    race_hash.output(dataset: 'race_summary_stratified' ) ; 

  END ; 

RUN ; 

 

/*SORT OUTPUT DATASET FOR READABILITY*/ 

PROC SORT DATA = race_summary_stratified ; 

  BY dp sex race imputedrace ; 

RUN ; 

The only lines of code that changed were the definekey and definedata steps when declaring the hash 
table to include the categorical variables for stratification.  



21 

The output is shown below: 

 

Output 17. HASH output dataset with stratifications 

 

PERL REGULAR EXPRESSIONS 

Regular expressions are typically considered when code needs to evaluate, deconstruct, or analyze text 
strings, but some basic regular expression functions can also be used to generate comparisons between 
variables. Due to how the comparisons are generated, this can be a preferred method if later steps will 
also need a counter variable, as a numeric variable is generated in this process that could be used for 
other procedures. 

In this initial example, PRXPARSE and PRXMATCH are used in conjunction. Combining these two 
provides variables that can be used in both further selection steps as well as in summing. 

PRXPARSE establishes the values that are being searched in the variable and the condition(s) for those 
values. PRXMATCH completes the search within the variable of interest. 

DATA REGEX_01 ; 

  SET dem_l3_catvars ( KEEP = DP HISPANIC IMPUTEDHISPANIC ) ; 

/* STARTING WITH RECORDS WHERE THE VALUE OF HISPANIC IS N = NO     */ 

  IF upcase ( HISPANIC ) in ('N') THEN DO ; 

 /* PRXPARSE COMPILES THE PERL REGULAR EXPRESSION FOR USE IN SEARCH */ 

 /* SETTING UP non-NO VALUES – Y = YES, U = UNKNOWN, AND BLANK      */ 

 /* THE i OPTION MAKES SEARCH CASE INSENSITIVE                      */ 

   patternID = prxparse ('/Y|U| /i') ; 

 /* PRXMATCH FINDS POSITION OF PATTERN MATCH IN COMPARED VARIABLE   */ 

   mismatch = prxmatch ( patternID , IMPUTEDHISPANIC ) ; 

   END ; 

 /* REPEAT TO CREATE COMPARISON FOR OTHER VALUES OF VARIABLES       */ 

   ELSE IF upcase ( HISPANIC ) in ('Y') THEN DO ; 

    patternID = prxparse ('/N|U| /i') ; 

 mismatch = prxmatch ( patternID , IMPUTEDHISPANIC ) ; 

   END ; 

   ELSE IF upcase ( HISPANIC ) in ('U') THEN DO ; 

    patternID = prxparse ('/Y|N| /i') ; 

 mismatch = prxmatch ( patternID , IMPUTEDHISPANIC ) ; 

   END ; 

RUN ; 

 

Here is a sample of the output dataset REGEX_01: 

 

 

Output 18. PERL regular expression output dataset  

The PRXPARSE function yields the values in patternID. This enumerates the combinations or “patterns” 
of search that were in the data. The PRXMATCH function yields the values in mismatch. This provides a 
numeric value indicating the position in the string of the parsed values with the number of that position or 
a non-match to the parsed values with a 0. For this comparison between a variable and its imputed 
version, the parsed search was based on looking for mismatches, so for this example the records where 
mismatch = 1 indicate the values are not the same; if the strings were longer, the numbers could have 
varied more. 

A sorted, deduplicated version of this dataset can make it easier to determine if the results of PRXPARSE 
could aid in further subsetting or crosstabulating the data based on the combinations of values in 
patternID. Here is one method to create that dataset using PROC SORT followed by its output: 

 

/* SHOW FINAL COMBINATIONS OF VALUES TO DETERMINE HOW TO BEST */ 

/* STRUCTURE SELECTION OF RECORDS FOR COUNTS                  */ 

PROC SORT DATA = REGEX_01  OUT = REGEX_01_DISTINCT NODUPKEY ; 

  BY _ALL_ ; 

RUN ; 

 

Output 19. De-duplicated PERL regular expression output dataset  

In this case, HISPANIC = N links to patternID = 1, HISPANIC = U links to patternID = 2, and HISPANIC = 
Y links to patternID = 3. Depending on what type of analysis is being done after the comparison, these 
values could be used in conjunction with the match or non-match (indicated in this case by mismatch), but 
it is important to see how the patterns were assigned, and a de-duplicated dataset makes that easy to 
visualize. 

If the only relevant information is the question of a match or non-match, this can be accomplished using 
only PRXMATCH.  

 

DATA REGEX_02 ; 

 SET dem_l3_catvars ( KEEP = DP HISPANIC IMPUTEDHISPANIC ) ; 

 /* IF DISTINCT IDENTIFICATION OF THE PATTERN IS NOT NEEDED */ 

 /* THE SEARCH CAN BE DONE USING ONLY PRXMATCH              */ 

  IF upcase ( HISPANIC ) in ('N') THEN DO ; 

   mismatch = prxmatch ( '/Y|U| /i' , IMPUTEDHISPANIC ) ; 

  END ; 
 /* REPEAT TO CREATE COMPARISON FOR OTHER VALUES OF VARIABLE */ 

  ELSE IF upcase ( HISPANIC ) in ('Y') THEN DO ; 

   mismatch = prxmatch ( '/N|U| /i' , IMPUTEDHISPANIC ) ; 

END ;  

RUN ;  

 

Note that this output has the same values as was generated in REGEX_01 but without the inclusion of 
position. 

 

Output 20. PERL regular expression output dataset using only PRXMATCH 

Particularly if dealing with large data or restrictions on available memory, try to be conservative in the 
construction of regular expressions as complex ones can become resource intensive. However, the use 
of regular expressions is often more efficient than longer or repeated procedures. 

PROC FREQ: ADDITONAL APPLICATIONS 

To further analyze the PROC COMPARE output, we can perform use PROC FREQ for counts of matches 
and mismatches. See the following code and output below for the race variable.  

This evaluates the PROC COMPARE output (shown in Output 4) for records where _type_ = ’DIF’: ‘.’ 
values match, and ‘X’ values are mismatches. The code can substitute the hispanic variable for race to 
produce a mismatch analysis for Hispanic/Imputed Hispanic. 

 

title1 'Mismatch Analysis of Proc Compare Output for Race/Imputed Race 

(stratified by DP and Sex)'; 

 

PROC FREQ DATA=comp_strat ( where = ( _type_ = 'DIF' ) ) ; 

  TABLES _type_ * dp * sex * race / NOROW NOCOL NOPERCENT ; 

RUN ; 

 

 

Output 21. PROC COMPARE output analyzed with PROC FREQ 

 

Hash table output can be further analyzed with PROC FREQ to produce frequencies of the matches vs 
mismatches: 

title1 'Frequencies of Race Matches'; 

 

PROC FREQ DATA = race_summary_stratified ; 

  TABLES race_match_count / NOROW NOCOL NOPERCENT ; 

  BY dp sex; 

RUN ; 

 

title1 'Frequencies of Race Misatches'; 

 

PROC FREQ DATA = race_summary_stratified ; 

  TABLES race_mismatch_count / NOROW NOCOL NOPERCENT ; 

  BY dp sex; 

RUN ; 

 

 

Output 22. HASH output analyzed with PROC FREQ 

Additionally, PROC FREQ can produce crosstabulations of matches vs mismatches using the hash table 
output dataset (shown in Output 17): 

 

 

 

 

 



27 

 

title1 'Crosstabulation of Race Matches'; 

 

PROC FREQ DATA = race_summary_stratified ; 

  TABLES dp*sex*race_match_count / NOROW NOCOL NOPERCENT ; 

RUN ; 

 

title1 'Frequencies of Race Mismatches'; 

 

PROC FREQ DATA = race_summary_stratified ; 

  TABLES dp*sex*race_mismatch_count / NOROW NOCOL NOPERCENT ; 

RUN ; 

 

 

Output 23. HASH output analyzed with PROC FREQ on matches and mismatches 

 

RESULTS 

The focus of these methods is on data that is not inherently numeric, even if it presents as numbers. By 
utilizing data in a character format, more procedures can be considered, and more flexibility is possible in 
the comparative analysis. These tools can help you to understand the quality of your data before research 
with it begins, can be a crucial part of analytical work, and can even provide post-analytical summaries. At 
any of these stages, decisions can be made to select or subset records based on the patterns or trends 
that are found. 

An initial exploration of the data is often helpful, and starting with PROC FREQ or PROC SQL would be 
productive.  Both offer an initial code block that is short to get started but can also be extensively modified 
or customized with available options.  This provides easy manipulation of the data to help examine initial 
trends and decide what details merit further exploration. 

Both PROC COMPARE and PROC MEANS can add additional levels of details and create more nuanced 
comparisons. Consider starting with them if the data is already familiar or if specifications already tightly 
define the types of comparisons that are needed.  

If the data requires reworking, then it is worth stepping directly into HASH tables or implementing the use 
of regular expressions to derive needed components and make comparisons.  Since either of these 
methods can then easily be combined with others, look at what the end result needs to be and build the 
series of procedures to meet that result. 

Should all else fail, default to PROC FREQ. It is a true workhorse in this context. 

CONCLUSION 

This paper explores various methods that can compare reported versus imputed values using SAS. As 
demonstrated, there is no one preferred method, no “one proc to rule them all”, as it depends on what is 
most important to consider in the analysis, whether that is calculating counts, mismatches, percentages, 
or considering efficiency while performing these analyses. The choice is driven by the nature of the data 
and the specific research questions to answer, so this paper can serve as a guide to help make those 
choices clearer.  

APPENDIX 

Below find the code which generated the fabricated source data set used in this paper: 

DATA dem_l3_catvars ; 

    /* SET THE NUMBER OF OBSERVATIONS DESIRED */ 

    DO i = 1 to 100; /* ADJUST NUMBER OF ROWS AS NEEDED */ 

        /* ASSIGN VALUES FOR DP */ 

        SELECT (floor(ranuni(123) * 3) + 1); 

            WHEN (1) DP = 'DP01' ; 

            WHEN (2) DP = 'DP02' ; 

            WHEN (3) DP = 'DP03' ; 

        END ; 

  

        /* GENERATE RANDOM COUNT VALUE; RANDOM NUMBERS 10 AND GREATER */ 

        count = floor(10 + ranuni(123) * 10000) ;  

        /* ASSIGN VALUES FOR HISPANIC */ 

        IF ranuni(123) < 0.33 THEN hispanic = 'Y' ; 

        ELSE IF ranuni(123) < 0.66 THEN hispanic = 'N' ; 

        ELSE hispanic = 'U' ; 

        /* ASSIGN VALUES FOR IMPUTEDHISPANIC;10% CHANCE OF BEING NULL */ 

        IF ranuni(123) < 0.1 THEN imputedhispanic = '' ;  

        ELSE IF ranuni(123) < 0.33 THEN imputedhispanic = 'Y' ; 

        ELSE IF ranuni(123) < 0.66 THEN imputedhispanic = 'N' ; 

        ELSE imputedhispanic = 'U' ; 

        /* ASSIGN VALUES FOR RACE; GENERATES VALUES 0 TO 6 */ 

        race = put(floor(ranuni(123) * 7), 1.) ;  

        IF ranuni(123) < 0.1 THEN race = 'M' ; /* 10% CHANCE OF BEING 'M' */ 

        /* ASSIGN VALUES FOR IMPUTEDRACE; 10% CHANCE OF BEING NULL */ 

        IF ranuni(123) < 0.1 THEN imputedrace = '' ;  

        ELSE imputedrace = put(floor(ranuni(123) * 7), 1.) ;  

        /* GENERATES VALUES 0 TO 6; 10% CHANCE OF BEING 'M' */ 

        if ranuni(123) < 0.1 then imputedrace = 'M' ;  

        /* ASSIGN VALUES FOR SEX */ 

        IF ranuni(123) < 0.5 THEN sex = 'M' ; 

        ELSE if ranuni(123) < 0.8 THEN sex = 'F' ; 

        ELSE sex = 'U' ; 

        OUTPUT ; /* WRITE RECORD TO DATASET */ 

    END ; 

    DROP i ; /* DROP LOOP INDEX VARIABLE */ 

RUN ; 


